# Maintainer: Justin Davis (juster) <jrcd83@gmail.com>
# $Id: PKGBUILD 86651 2013-03-20 22:59:24Z eric $

pkgname=perl-cpan-perl-releases
pkgver=1.06
pkgrel=1
pkgdesc='Mapping Perl releases on CPAN to the location of the tarballs'
arch=(any)
license=(PerlArtistic GPL)
options=(!emptydirs)
depends=('perl>=5.6.0')
url="https://metacpan.org/release/CPAN-Perl-Releases"
source=("http://search.cpan.org/CPAN/authors/id/B/BI/BINGOS/CPAN-Perl-Releases-$pkgver.tar.gz")
md5sums=('e66679b29edcab32f68bb6c4978d51c5')

build()
(
  cd CPAN-Perl-Releases-$pkgver
  export PERL_MM_USE_DEFAULT=1 PERL_AUTOINSTALL=--skipdeps
  unset PERL5LIB PERL_MM_OPT
  /usr/bin/perl Makefile.PL
  make
)

check()
(
  cd CPAN-Perl-Releases-$pkgver
  export PERL_MM_USE_DEFAULT=1
  unset PERL5LIB
  make test
)

package()
(
  cd CPAN-Perl-Releases-$pkgver
  make install INSTALLDIRS=vendor DESTDIR="$pkgdir"
)

# Local Variables:
# mode: shell-script
# sh-basic-offset: 2
# End:
# vim:set ts=2 sw=2 et:
